# $Id: PKGBUILD 178978 2013-03-01 10:23:20Z andrea $
# Maintainer:
# Contributor: Pierre Schmitz <pierre@archlinux.de>

pkgname=qca-gnupg
pkgver=2.0.0
_pkgver=2.0.0-beta3
pkgrel=5
pkgdesc="Qt Cryptographic Architecture"
arch=('i686' 'x86_64')
url="http://delta.affinix.com/qca/"
license=('LGPL')
depends=('qca')
optdepends=('gnupg' 'gnupg2')
source=("http://delta.affinix.com/download/qca/2.0/plugins/${pkgname}-${_pkgver}.tar.bz2")
md5sums=('9b4d020efd835a52d98b2ced9ae79c4b')

build() {
	cd "${srcdir}"/${pkgname}-${_pkgver}
	./configure \
		--release \
		--no-separate-debug-info
	make
}

package() {
	cd "${srcdir}"/${pkgname}-${_pkgver}
	make INSTALL_ROOT="${pkgdir}" install
}
